In and Out

Edited by Paul Abbott

In and Out offers readers an opportunity to ask questions of the experts. *The Mathematica Journal* encourages readers to submit problems in care of the editor. Answers posted to the *Mathematica* newsgroup, comp.soft-sys.math.mathematica, that appear here are edited for clarity and length.

■ Working with Intervals

Q: **A list of times of the form $\{\{t_1, t_1 + \delta t_1\}, \{t_2, t_2 + \delta t_2\}, \ldots, \{t_n, t_n + \delta t_n\}\}$ represents the periods $\{t_i, t_i + \delta t_i\}$ when a particular system is not working. For two such lists, how can I determine the periods when *both* systems are not working?**

A: Carl Woll (carlw@wolfram.com) answers: The function **timelist** simulates non-overlapping timing data of the form $\{\{t_1, t_1 + \delta t_1\}, \{t_2, t_2 + \delta t_2\}, \ldots, \{t_n, t_n + \delta t_n\}\}$.

```
In[1]:= timelist[n_] := Function[{a, b}, {a, a + (b - a) Random[]}] @@@
          Partition[Sort[Array[Random[] &, n]], 2, 1]
```

Here are two lists of timing data.

```
In[2]:= list1 = timelist[7]
```

Out[2]=

$$\begin{pmatrix} 0.266593 & 0.347591 \\ 0.421387 & 0.508681 \\ 0.53448 & 0.544166 \\ 0.595902 & 0.606203 \\ 0.697767 & 0.703369 \\ 0.74441 & 0.759786 \end{pmatrix}$$

In[3]:= **list2 = timelist[9]**

Out[3]=

$$\begin{pmatrix} 0.0289648 & 0.0884388 \\ 0.10387 & 0.120679 \\ 0.34196 & 0.345286 \\ 0.406686 & 0.408005 \\ 0.497661 & 0.512154 \\ 0.55351 & 0.656577 \\ 0.712268 & 0.777817 \\ 0.795792 & 0.805201 \end{pmatrix}$$

After converting each period $\{t_i, t_i + \delta t_i\}$ into an **Interval** using **Apply**, **Interval**·. **Intersection** returns those periods when both systems are not working.

In[4]:= **IntervalIntersection[Interval @@ list1, Interval @@ list2]**

Out[4]= Interval[{0.34196, 0.345286}, {0.497661, 0.508681},
{0.595902, 0.606203}, {0.74441, 0.759786}]

■ HamiltonianCycle

Q: I am interested in counting the number of paths from a node in a network back to that node, counting loops (cycles) only once. Is there any readily available code that does this?

A: Steve Luttrell (steve_usenet@luttrell.org.uk) answers: Use **HamiltonianCycle** in the **DiscreteMath`Combinatorica`** package. Here is an example of how this solves your problem.

Load the **DiscreteMath`Combinatorica`** package.

In[1]:= **<< DiscreteMath`Combinatorica`**

Here is a random, labeled graph on 10 vertices with an edge probability of 0.5.

In[2]:= **ShowGraph[*g* = RandomGraph[10, 0.5]]**



Using **Highlight**, display a **HamiltonianCycle** of the graph **g**.

In[3]:= **ShowGraph[Highlight[*g*, {Partition[HamiltonianCycle[*g*], 2, 1]}]]**



The total number of Hamiltonian cycles is equal to the number of loops that start from and eventually return to any starting node, having visited every other node exactly once.

In[4]:= **Length[HamiltonianCycle[*g*, All]]**

Out[4]= 52

■ Listable Subvalues

Q: I am writing functions for translating graphics primitives. The following function translates a single line object.

```
In[1]:= f[vec_][Line[a_]] := Line[(vec + # &) /@ a]

In[2]:= b = Line[{{1, 2}, {3, 4}, {5, 6}}];

In[3]:= f[{1, 3}][b]
```

Out[3]= $\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right]$

However, if I have a list of lines it does not work.

```
In[4]:= c = {b, Line[{{2, 3}, {5, 6}, {9, 3}}]};

In[5]:= f[{1, 3}][c]
```

Out[5]= $f(\{1, 3\})\left(\left\{\text{Line}\left[\begin{pmatrix} 1 & 2 \\ 3 & 4 \\ 5 & 6 \end{pmatrix}\right], \text{Line}\left[\begin{pmatrix} 2 & 3 \\ 5 & 6 \\ 9 & 3 \end{pmatrix}\right]\right\}\right)$

What is the best solution?

A: You can map your function over the list of lines.

```
In[6]:= f[{1, 3}] /@ c
```

Out[6]= $\left\{\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right], \text{Line}\left[\begin{pmatrix} 3 & 6 \\ 6 & 9 \\ 10 & 6 \end{pmatrix}\right]\right\}$

Adding a rule to ***f*** for the case of list arguments makes this mapping operation automatic.

```
In[7]:= f[vec_][a_List] := f[vec] /@ a

In[8]:= f[{1, 3}][c]
```

Out[8]= $\left\{\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right], \text{Line}\left[\begin{pmatrix} 3 & 6 \\ 6 & 9 \\ 10 & 6 \end{pmatrix}\right]\right\}$

David Park (djmp@earthlink.net) provides an alternative solution: define ***f*[vec_]** as **Function**. The third argument to **Function** is a list of attributes for the purpose of evaluation.

```
In[9]:= Clear[f];
```

```
In[10]:= f[vec_] := Function[{a}, Line[(vec + # &) /@ First[a]], {Listable}]

In[11]:= f[{1, 3}][b]
```

Out[11]= $\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right]$

```
In[12]:= f[{1, 3}][c]
```

Out[12]= $\left\{\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right], \text{Line}\left[\begin{pmatrix} 3 & 6 \\ 6 & 9 \\ 10 & 6 \end{pmatrix}\right]\right\}$

Functions for transforming objects in 2 and 3 dimensions are defined in [1] with the packages available at [2].

■ $f(x) = f(2\,x) + f(2\,x + 1)$

Q: How can I solve the functional equation $f(x) = f(2\,x) + f(2\,x + 1)$?

A: The solution to this functional equation is given at mathworld.wolfram.com/FunctionalEquation.html. Noting that

In[1]:= $\text{Simplify}\left[1 + \frac{1}{x} == \left(1 + \frac{1}{2\,x}\right)\left(1 + \frac{1}{2\,x + 1}\right)\right]$

Out[1]= True

then taking logs of both sides, one sees that $f(x) = c \log(1 + 1/x)$, where c is arbitrary, satisfies the functional equation. More generally, since

In[2]:= $1 + \frac{1}{x} == \prod_{i=1}^{n}\left(1 + \frac{1}{n\,x + i - 1}\right)$ // FullSimplify

Out[2]= True

we observe that $f(x) = c \log(1 + 1/x)$ is a solution to the functional equation

$$f(x) = \sum_{i=1}^{n} f(n\,x + i - 1)$$

for $n = 2, 3, \ldots$.

One way to solve the functional equation is to assume that the asymptotic behavior of the solution is $f(x) \sim c_1/x + c_2/x^2 + \cdots$.

In[3]:= $f_{m\_}(x\_) := \sum_{n=1}^{m} \frac{c_n}{x^n}$

Substitute this sum into the functional equation and expand into an asymptotic series.

In[4]:= **Series[$f(x) - (f(2x) + f(2x+1))$ /. $f \to f_6$, {x, ∞, 6}]**

Out[4]= $$\left(\frac{c_1}{4}+\frac{c_2}{2}\right)\left(\frac{1}{x}\right)^2+\left(-\frac{c_1}{8}+\frac{c_2}{4}+\frac{3c_3}{4}\right)\left(\frac{1}{x}\right)^3+\left(\frac{c_1}{16}-\frac{3c_2}{16}+\frac{3c_3}{16}+\frac{7c_4}{8}\right)\left(\frac{1}{x}\right)^4+\left(-\frac{c_1}{32}+\frac{c_2}{8}-\frac{3c_3}{16}+\frac{c_4}{8}+\frac{15c_5}{16}\right)\left(\frac{1}{x}\right)^5+\left(\frac{c_1}{64}-\frac{5c_2}{64}+\frac{5c_3}{32}-\frac{5c_4}{32}+\frac{5c_5}{64}+\frac{31c_6}{32}\right)\left(\frac{1}{x}\right)^6+O\left(\left(\frac{1}{x}\right)^7\right)$$

Solving for the coefficients α_n, one sees that $c_n = (-1)^{n+1}\, c_1 / n$.

In[5]:= **Solve[% == 0, Table[c_n, {n, 2, 6}]]**

Out[5]= $$\left\{\left\{c_2 \to -\frac{c_1}{2},\ c_3 \to \frac{c_1}{3},\ c_4 \to -\frac{c_1}{4},\ c_5 \to \frac{c_1}{5},\ c_6 \to -\frac{c_1}{6}\right\}\right\}$$

Summing the asymptotic series, one obtains the same solution as earlier, $f(x) = c_1 \log(1 + 1/x)$.

In[6]:= $$\sum_{n=1}^{\infty} \frac{(-1)^{n+1} c_1}{n\, x^n}$$

Out[6]= $$\log\left(1+\frac{1}{x}\right) c_1$$

Here we apply the same method to $g(x) = g(2x + a) + g(2x + b)$.

In[7]:= **Solve[Series[$g(x) -$ ($g(2x+a) + g(2x+b)$) /. $g \to f_4$, {x, ∞, 4}] == 0,
Table[c_n, {n, 2, 4}]] // Simplify**

Out[7]= $$\left\{\left\{c_4 \to -\frac{1}{4}(a^3+b\,a^2+b^2\,a+b^3)\,c_1,\ c_3 \to \frac{1}{3}(a^2+b\,a+b^2)\,c_1,\ c_2 \to -\frac{1}{2}(a+b)\,c_1\right\}\right\}$$

The pattern of the coefficients is clear: the coefficient c_{n+1} is $(-1)^n\, c_1 \sum_{i=0}^{n} a^i\, b^{n-i} / (n+1)$. Summing the asymptotic series, one obtains the general solution.

In[8]:= **Simplify**$$\left[\sum_{n=0}^{\infty} \frac{(-1)^n \sum_{i=0}^{n} a^i\, b^{n-i}\, c_1}{(n+1)\, x^{n+1}}\right]$$

Out[8]= $$\frac{\left(\log\left(\frac{a+x}{x}\right) - \log\left(\frac{b+x}{x}\right)\right) c_1}{a-b}$$

Absorbing the factor $(a-b)$ into the arbitrary constant, the solution can be written as $g(x) = c \log\left(\frac{a+x}{b+x}\right)$. This solution results by taking the logarithm of the following identity.

In[9]:= $\left(\frac{a+x}{b+x}\right) = \left(\frac{a+2\,x+a}{b+2\,x+a}\right)\left(\frac{a+2\,x+b}{b+2\,x+b}\right)$ **// Simplify**

Out[9]= True

Finally, we apply the same method to $h(x) = h(x+1) + h(x^2 + x + 1)$.

In[10]:= **Solve[Series[$h(x) - (h(x+1) + h(x^2 + x + 1))$ /. $h \to f_{10}$, {x, ∞, 11}] == 0, Table[c_n, {n, 2, 10}]]**

Out[10]= $\left\{\left\{c_2 \to 0,\ c_3 \to -\frac{c_1}{3},\ c_4 \to 0,\ c_5 \to \frac{c_1}{5},\ c_6 \to 0,\ c_7 \to -\frac{c_1}{7},\ c_8 \to 0,\ c_9 \to \frac{c_1}{9},\ c_{10} \to 0\right\}\right\}$

The pattern of the coefficients is clear: the even coefficients vanish, $c_{2\,n} = 0$, and the odd coefficients read $c_{2\,n+1} = (-1)^n\, c_1 / (2\,n+1)$. Summing the asymptotic series, one obtains the general solution.

In[11]:= **Sum**$\left[\frac{(-1)^{\frac{n-1}{2}}\, c_1}{n\, x^n}, \{n, 1, \infty, 2\}\right]$

Out[11]= $\tan^{-1}\left(\frac{1}{x}\right) c_1$

See also [3], [4], and [5].

■ EventLocator

Q: The solution to the differential equation $y'(x) = \cos(x)$ with $y(0) = 0$ is $\sin(x)$.

In[1]:= **DSolve[{$y'(x) = \cos(x)$, $y(0) = 0$}, $y(x)$, x]**

Out[1]= {{$y(x) \to \sin(x)$}}

Solving the same equation numerically, I tried using the event function $y(x) - 1$ to find the extrema of $y(x)$.

In[2]:= **NDSolve[{$y'(x) = \cos(x)$, $y(0) = 0$}, y, {x, 0, 20}, Method → {"EventLocator", "Event" → $y(x) - 1$, "EventAction" :→ Sow[x]}] // Reap**

Out[2]= {{{y → InterpolatingFunction[(0. 20.), <>]}}, {}}

How can I find those points where $y(x) = 1$?

A: Mark Sofroniou (marks@wolfram.com) answers: An event is located when a change in sign in the event function is detected. For the function $\sin(x) - 1$, the sign is practically always negative and the chance of hitting zero is infinitesimal.

You can construct an appropriate event function for extrema, where $y'(x) = 0$, by noting that $y'(x) = \cos(x)$. Then it is possible to use the **Direction** option to restrict the detection to points corresponding to a maximum.

In[3]:=
```
NDSolve[{y′(x) == cos(x), y(0) == 0}, y,
    {x, 0, 20}, Method → {EventLocator, "Event" → cos(x),
        "EventAction" :→ Sow[x], "Direction" → −1}] // Reap
```

Out[3]=
$$\begin{pmatrix} \{y \to \text{InterpolatingFunction}[(0. \quad 20.), <>]\} \\ \{1.5708, 7.85398, 14.1372\} \end{pmatrix}$$

■ Operational Solutions to Differential Equations

Q: If $\phi(z)$ is the electrostatic potential on the axis of a cylindrically symmetric system, then the potential at the point (ρ, z), where ρ is the perpendicular distance from the axis, is given by the following (see, e.g., [6])

$$\phi(\rho, z) = \mathcal{J}_0\left(\rho \frac{\partial}{\partial z}\right)\phi(z).$$

How can I implement the operator $\mathcal{J}_0\left(\rho \frac{\partial}{\partial z}\right)$?

A: The definition of $\mathcal{J}_\nu(z)$ is given at functions.wolfram.com/03.01.02.0001.01.

In[1]:=
$$\mathcal{J}_\nu(z) == \sum_{k=0}^{\infty} \frac{(-1)^k \left(\frac{z}{2}\right)^{2k+\nu}}{\Gamma(k+\nu+1)\, k!}$$

Out[1]= True

Then, formally, one has

$$\mathcal{J}_0\left(\rho \frac{\partial}{\partial z}\right) \equiv \sum_{k=0}^{\infty} \frac{(-1)^k \left(\rho \frac{\partial}{\partial z}\right)^{2k}}{4^k\, k!^2} = \sum_{k=0}^{\infty} \frac{1}{k!^2}\left(-\frac{\rho^2}{4}\frac{\partial^2}{\partial z^2}\right)^k.$$

For an arbitrary potential $\phi(z)$, the operator formalism can be used to obtain the Taylor series expansion of $\phi(\rho, z)$ about $\rho = 0$ using **NestList**. For example, here are the first four terms.

In[2]:=
$$k = 0;\ \phi_4(\rho\_, z\_) = \text{Tr}@\text{NestList}\left[\left(++k;\ -\frac{\rho^2}{4k^2}\,\partial_{z,z}\,\#\right)\&,\ \phi(z),\ 3\right]$$

Out[2]=
$$-\frac{\phi^{(6)}(z)\,\rho^6}{2304} + \frac{1}{64}\phi^{(4)}(z)\,\rho^4 - \frac{1}{4}\phi''(z)\,\rho^2 + \phi(z)$$

For an axially symmetric potential, the Laplacian in cylindrical coordinates reads,

$$\nabla^2 \equiv \Delta = \frac{1}{\rho}\left(\frac{\partial}{\partial\rho}\left(\rho\,\frac{\partial}{\partial\rho}\right)\right) + \frac{\partial^2}{\partial z^2},$$

and the potential satisfies Laplace's equation, $\Delta V(\rho, z) = 0$. Verifying that the operator expansion produces a formal power-series solution is immediate.

In[3]:= $\dfrac{\partial^2 \phi_4(\rho, z)}{\partial z^2} + O[\rho]^6$

Out[3]= $\phi''(z) - \frac{1}{4}\,\phi^{(4)}(z)\,\rho^2 + \frac{1}{64}\,\phi^{(6)}(z)\,\rho^4 + O(\rho^6)$

In[4]:= $\text{Collect}\left[\frac{1}{\rho}\left(\partial_\rho\left(\rho\,\frac{\partial \phi_4(\rho, z)}{\partial \rho}\right)\right), \rho, \text{Simplify}\right]$

Out[4]= $-\frac{1}{64}\,\phi^{(6)}(z)\,\rho^4 + \frac{1}{4}\,\phi^{(4)}(z)\,\rho^2 - \phi''(z)$

In[5]:= % + %%

Out[5]= $O(\rho^6)$

Now for a concrete example. For a disk of radius R, with uniform surface charge density σ, oriented with its normal vector along the z-axis, here is the potential at $(0, 0, z)$.

In[6]:= $V_{R\_}(z\_) = \text{Assuming}\left[z > R > 0, \frac{\sigma}{4\pi\epsilon_0}\int_0^{2\pi}\int_0^R \frac{1}{\sqrt{s^2 + z^2}}\, s\, ds\, d\phi\right]$

Out[6]= $\dfrac{\left(\sqrt{R^2 + z^2} - z\right)\sigma}{2\,\epsilon_0}$

Using the operator formalism one obtains

In[7]:= $k = 0;\ V_{3,R}(\rho\_, z\_) = \text{Simplify /@ Tr @ NestList}\left[\left(++k;\ -\frac{\rho^2}{4k^2}\,\partial_{z,z}\,\#\right)\&,\ V_R(z),\ 2\right]$

Out[7]= $-\dfrac{3\,(R^4 - 4R^2 z^2)\,\sigma\,\rho^4}{128\,(R^2 + z^2)^{7/2}\,\epsilon_0} - \dfrac{R^2\,\sigma\,\rho^2}{8\,(R^2 + z^2)^{3/2}\,\epsilon_0} + \dfrac{\left(\sqrt{R^2 + z^2} - z\right)\sigma}{2\,\epsilon_0}$

There is another approach to this problem: In the case of azimuthal symmetry, the general solution to Laplace's equation $\Delta V = 0$ is (the multipole expansion),

$$V(r, \theta) = \sum_{l=0}^{\infty}\left(a_l\, r^l + \frac{b_l}{r^{l+1}}\right) P_l(\cos(\theta)), \tag{1}$$

where r and θ are the radial and polar spherical coordinates, respectively. Here is the truncated solution.

In[8]:= $V_{n\_}(r\_, \theta\_) := \sum_{l=0}^{n}\left(a_l\, r^l + \frac{b_l}{r^{l+1}}\right) P_l(\cos(\theta))$

Now, if the potential is known on the axis, that is $V(r,\,0) \equiv V(z)$, then one can use equation (1) to determine a_l and b_l by series expansion of $V(z)$ and term-by-term comparison. For $R > r > 0$, here is the series expansion of the axial potential.

In[9]:= $\mathbf{Simplify[V_R(r) + O[r]^5,\ R > 0]}$

Out[9]= $$\frac{R\,\sigma}{2\,\epsilon_0} - \frac{\sigma\,r}{2\,\epsilon_0} + \frac{\sigma\,r^2}{4\,R\,\epsilon_0} - \frac{\sigma\,r^4}{16\,(R^3\,\epsilon_0)} + O(r^5)$$

Now, equate this to $V(r,\,0)$ and solve.

In[10]:= $\mathbf{V_4(r,\,0) + O[r]^5}$

Out[10]= $$\frac{b_4}{r^5} + \frac{b_3}{r^4} + \frac{b_2}{r^3} + \frac{b_1}{r^2} + \frac{b_0}{r} + a_0 + a_1\,r + a_2\,r^2 + a_3\,r^3 + a_4\,r^4 + O(r^5)$$

In[11]:= $\mathbf{Solve[\% == \%\%,\ Join[Table[a_l,\ \{l,\,0,\,4\}],\ Table[b_l,\ \{l,\,0,\,4\}]]]}$

Out[11]= $$\Big\{\Big\{a_3 \to 0,\ b_0 \to 0,\ b_1 \to 0,\ b_2 \to 0,\ b_3 \to 0,\ b_4 \to 0,\ a_0 \to \frac{R\,\sigma}{2\,\epsilon_0},\ a_1 \to -\frac{\sigma}{2\,\epsilon_0},\ a_2 \to \frac{\sigma}{4\,R\,\epsilon_0},\ a_4 \to -\frac{\sigma}{16\,R^3\,\epsilon_0}\Big\}\Big\}$$

Hence we obtain the (truncated series expansion of the) potential of the disk off the axis in spherical coordinates.

In[12]:= $\mathbf{V_4(r,\,\theta)\ /.\ First[\%]}$

Out[12]= $$-\frac{\sigma\,(35\cos^4(\theta) - 30\cos^2(\theta) + 3)\,r^4}{128\,R^3\,\epsilon_0} + \frac{\sigma\,(3\cos^2(\theta) - 1)\,r^2}{8\,R\,\epsilon_0} - \frac{\sigma\cos(\theta)\,r}{2\,\epsilon_0} + \frac{R\,\sigma}{2\,\epsilon_0}$$

To compare this solution to that obtained earlier, we expand $\sqrt{R^2 + r^2} = R\sqrt{1 + (r/R)^2}$ into a series in r/R, valid for $0 < r < R$. See functions.wolfram.com/01.01.06.0002.01 and functions.wolfram.com/01.01.06.0003.01.

In[13]:= $$\sqrt{\mathbf{1 + x}} = {}_1F_0\Big(-\frac{1}{2};\,;\,-x\Big) = \sum_{n=0}^{\infty} \frac{(-1)^n\,\big(-\frac{1}{2}\big)_n}{n!}\,x^n$$

Out[13]= True

Check that we have the correct expansion for $V_R(r)$.

In[14]:= $$V_R(r) = \mathbf{FullSimplify}\Big[\frac{\sigma}{2\,\epsilon_0}\Big(R\sum_{n=0}^{\infty} \frac{(-1)^n\,\big(-\frac{1}{2}\big)_n}{n!}\,\Big(\frac{r}{R}\Big)^{2\,n} - r\Big),\ 0 < r < R\Big]$$

Out[14]= True

Hence the potential of the disk off the axis is given by

$$\frac{\sigma}{2\,\epsilon_0}\Big(R\sum_{n=0}^{\infty} \frac{(-1)^n\,(-\frac{1}{2})_n}{n!}\,\Big(\frac{r}{R}\Big)^{2\,n} P_{2\,n}(\cos(\theta)) - r\cos(\theta)\Big).$$

Changing coordinates from polar to cylindrical coordinates, $z = r\cos(\theta)$ and $r^2 = z^2 + \rho^2$, we verify that the two expansions are consistent.

In[15]:= $$\textbf{Simplify}\Big[R\sum_{n=0}^{3}\frac{(-1)^n\left(-\frac{1}{2}\right)_n\left(\frac{r}{R}\right)^{2n}P_{2n}(\cos(\theta))}{n!} - r\cos(\theta)\ /.\ \cos(\theta)\to\frac{z}{r}\ /.\ r\to\sqrt{z^2+\rho^2}\Big] + O[\rho]^5$$

Out[15]= $$\left(\frac{z^6}{16R^5}-\frac{z^4}{8R^3}+\frac{z^2}{2R}-z+R\right)+\left(-\frac{15z^4}{32R^5}+\frac{3z^2}{8R^3}-\frac{1}{4R}\right)\rho^2+\left(\frac{45z^2}{128R^5}-\frac{3}{64R^3}\right)\rho^4+O(\rho^5)$$

In[16]:= $$\textbf{Simplify}\Big[\textbf{Series}\Big[\% - \frac{2\epsilon_0}{\sigma}V_{3,R}(\rho, z), \{z, 0, 3\}\Big], R > 0\Big]\ //\ \textbf{Normal}$$

Out[16]= 0

Elmar Zeitler (zeitler@fhi-berlin.mpg.de) submitted another example of an operator expansion. Using the integral definition (functions.wolfram.com/03.01.07.0005.01),

$$J_n(z) = \frac{(-i)^n}{\pi}\int_0^\pi e^{i z\cos(\theta)}\cos(n\,\theta)\,d\theta,$$

and the identity (functions.wolfram.com/01.07.16.0096.01),

$$\cos(n\,\theta) = T_n(\cos(\theta)),$$

then the change of variables $\cos(\theta)\to x$ yields

$$J_n(z) = \frac{(-i)^n}{\pi}\int_{-1}^1\frac{e^{i z x}}{\sqrt{1-x^2}}\,T_n(x)\,dx.$$

Now, $T_n(x)$ is a polynomial in x and, since $x^k\,e^{i z x} = (-i)^k\,\partial_{\{z,k\}}\,e^{i z x}$, we see that

$$J_n(z) = \frac{(-i)^n}{\pi}\,T_n(-i\,\partial_z)\int_{-1}^1\frac{e^{i z x}}{\sqrt{1-x^2}}\,dx = (-i)^n\,T_n(-i\,\partial_z)\,J_0(z).$$

Implementation of this operator expansion is direct.

In[17]:= $\textbf{Table}[\{n, (-i)^n\,(\textbf{Expand}[z\,T_n(-i\,z)]\ /.\ z^{k\_\cdot} :> \partial_{\{z,k-1\}}\#)\ \&[J_0(z)]\ //\ \textbf{Simplify}\}, \{n, 0, 4\}]$

Out[17]=

0	$J_0(z)$
1	$J_1(z)$
2	$J_2(z)$
3	$J_3(z)$
4	$J_4(z)$

■ Cluster Analysis

Q: For an arbitrary matrix of non-negative integers, how can I obtain the sum of those matrix elements that are surrounded by zeros? As a concrete example, for the following matrix an output of {4, 5, 7, 5, 4} is required.

In[1]:= $$\mathbf{mat} = \begin{pmatrix} 0 & 0 & 0 & 1 & 0 & 0 & 0 & 5 & 0 & 0 \\ 0 & 0 & 0 & 3 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 1 & 3 & 0 & 0 \\ 1 & 2 & 2 & 0 & 0 & 0 & 1 & 0 & 0 & 3 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix};$$

The order in which the groups surrounded by zero is summed does not matter.

A: Carl Woll (carlw@wolfram.com) answers: First, use **SparseArray** to get the positions of non-zero elements.

In[2]:=
```
datarules = Most[ArrayRules[SparseArray[mat]]]
```

Out[2]=
```
{{1, 4} → 1, {1, 8} → 5, {2, 4} → 3, {3, 2} → 1, {3, 7} → 1, {3, 8} → 3, {4, 1} → 1,
  {4, 2} → 2, {4, 3} → 2, {4, 7} → 1, {4, 10} → 3, {5, 3} → 1, {5, 10} → 1}
```

Next, define a distance function yielding 0 for identical elements, 1 for adjacent elements, and a big number, say 10, for nonadjacent elements.

In[3]:=
```
myDistance[d1_, d2_] := Clip[Total[|d1 − d2|], {0, 1}, {10, 10}]
```

Then, load the **Statistics`ClusterAnalysis`** package and use **FindClusters** with the **Agglomerate** method.

In[4]:=
```
Needs["Statistics`ClusterAnalysis`"]
```

In[5]:=
```
FindClusters[datarules, Method → Agglomerate,
  DistanceFunction → myDistance]
```

Out[5]=
```
{{1, 3}, {5}, {1, 1, 2, 2, 1}, {1, 3, 1}, {3, 1}}
```

Finally, total the cluster values.

In[6]:=
```
Total /@ %
```

Out[6]=
```
{4, 5, 7, 5, 4}
```

Here is a function to do all the steps.

In[7]:=
```
ClusterSums[array_] := Module[{datarules, clusteredvalues},
    datarules = Most[ArrayRules[SparseArray[array]]];
    clusteredvalues = FindClusters[datarules,
        Method → Agglomerate, DistanceFunction → myDistance];
    Total /@ clusteredvalues]
```

Check that **ClusterSums** works on **mat**.

```
In[8]:= ClusterSums[mat]

Out[8]= {4, 5, 7, 5, 4}
```

As a bonus, this approach can be extended to handle higher dimensional arrays.

■ References

[1] T. Wickham-Jones, *Mathematica Graphics: Techniques & Applications*, New York: TELOS/Springer-Verlag, 1994.

[2] library.wolfram.com/infocenter/Books/3753.

[3] E. Castillo, J. M. Gutiérrez, and A. Iglesias, "Solving a Functional Equation," *The Mathematica Journal*, **5**(1), 1995 pp. 82–86.

[4] library.wolfram.com/infocenter/Articles/857.

[5] library.wolfram.com/infocenter/Conferences/4760.

[6] E. Zeitler, "Operators in Electron Optics," *Optik* **3**, 1995 pp. 103–134.

Paul Abbott
School of Physics, M013
The University of Western Australia
35 Stirling Highway
Crawley WA 6009, Australia
tmj@physics.uwa.edu.au
physics.uwa.edu.au/~paul